# $Id: PKGBUILD 82310 2013-01-14 20:58:57Z thestinger $
# Maintainer: Daniel Micay <danielmicay@gmail.com>
# Contributor: Aaron Schaefer <aaron@elasticdog.com>
# Contributor: Daniel J Griffiths <ghost1227@archlinux.us>

pkgname=vim-surround
_sourceid=12566
pkgver=2.0
pkgrel=2
pkgdesc='Provides mappings to easily delete, change, and add paired "surroundings"'
arch=('any')
url='http://www.vim.org/scripts/script.php?script_id=1697'
license=('custom:vim')
groups=('vim-plugins')
depends=('vim')
install=vimdoc.install
source=("http://www.vim.org/scripts/download_script.php?src_id=$_sourceid"
        'license.txt')
md5sums=('c3d5129befb14ae82c1e7769253b703c'
         'cd49910a3dde6fe0a7e24ebe8cbf7066')

package() {
  cd "$srcdir"

  local installpath="$pkgdir/usr/share/vim/vimfiles"

  install -Dm644 doc/surround.txt "$installpath/doc/surround.txt"
  install -Dm644 plugin/surround.vim "$installpath/plugin/surround.vim"
  install -Dm644 license.txt "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}
